
Planners Lab Keywords
AVERAGE

The AVERAGE function returns the average for the specified range of columns. The AVERAGE function uses the THRU keyword to specify the range of columns to be used in it.

Syntax

AVERAGE(COL<column name> THRU COL<column name>)

Example

average = AVERAGE(COL 2006 THRU COL 2009)

BCRATIO

The BCRATIO function computes the ratio of total discounted cash inflows to the total discounted cash outflows.

Syntax

BCRATIO (Cash In, Cash Out, Discount Rate)

Example

Cash in= 0, 45000, 76000, 108000

Cash out = 0, 15000, 26000, 28000

Discount rate = .15

Benefit cost ratio = BCRATIO (Cash in, Cash out, Discount rate)

[RESULTS] = NaN, 3, 2.95, 3.28

CEILING

The CEILING function returns the closest whole number that is greater than or equal to the number. In other words, this function takes the given variable and rounds it up to the nearest whole number.

Syntax

CEILING (<variable name>)

Example

numbers = 1.7, 2.3, 4.5, 6.4

ceiling = CEILING (numbers)

[RESULTS] = 2, 3, 5, 7

COL

The COL keyword is used in special column equations while referring to and performing the operations on the Column Statements or to access a particular column in an equation.

Syntax

COL <column name>

Example

Year = SUM(COL Jan THRU COL Dec)

COMMENT

The COMMENT keyword is used in Planners Lab to enable the user to insert comments about the model. All the comments start with the keyword “COMMENT” and all following content is ignored by the Planners Lab until the start of next equation (line).

Syntax

COMMENT <content of the comment>

Example

COMMENT this node in the model is about marketing division employee details and also the work hours and the pay per hour

FLOOR

The FLOOR function returns the closest whole number that is less than or equal to the number. In other words, this function takes the given variable and rounds it down to the nearest whole number.

Syntax

FLOOR (<variable name>)

Example

numbers = 1.7, 2.3, 4.5, 6.4

floor = FLOOR (numbers)

[RESULTS] = 1, 2, 4, 6

FOR

The FOR keyword can be used when the same value is to be repeated for a certain number of columns. Two values have to be specified for the FOR keyword:

1. the value to be repeated and

2. the number of times to repeat.

Both value and times can be any expression, but it is recommended to make times a number.

Syntax

<value> FOR <time periods>

Example

Sales = 20000 FOR 3, 40000

OR

Sales = (1000 * 2) FOR 3, 5000

NOT VALID

(2 FOR 3) FOR 4

FORECAST

The FORECAST function forecasts the value of a dependent variable from an independent variable by using linear regression.  In the equation, at least two values must be provided for the variable before using the FORECAST function. The function calculates the value of the variable in future columns based on a linear representation of its previous relationship with an independent variable in the earlier columns. Hence, the relationship is determined by comparing the values of expressions in the earlier columns of this variable with the corresponding column values of the independent variable.

Syntax

<variable name> = <expression>, <expression>, <....>, FORECAST(independent variable)

Example

Unit price = 295, 199, 240, 310

Units sold = 47000, 59000, 50700, FORECAST(Unit price)

[RESULTS] = 47000, 59000, 50700, 44275.26

FUTURE

The FUTURE keyword is used in an equation when the value of the variable in one column is dependent on the value of an independent variable in a future column(s). A column offset can be specified how far in the future to retrieve a value. If no column offset is specified then 1 is used. If no column exists the value will be NaN.

Syntax

FUTURE <column offset> <variable name>

Example

Units Sold = 345, 450, 498, 520

Units Order = FUTURE Units Sold

OR

Units Order = FUTURE 2 Units Sold

IF .. THEN .. ELSE ..

The IF .. THEN .. ELSE .. keywords form a conditional expression and are used for a conditional selection for a variable value. The value of the variable is determined based on given conditions. If the “condition” is true then “true-expression” is executed and if it is false “false- expression” is executed. The condition must be of the following format:

<expression 1> <comparator> <expression 2>

Multiple condition can be specified using a conditions operator (AND & OR). A multiple condition must be of the following format:

<condition 1> <conditional operator> <condition 2>

The true-expression, false-expression, expression 1 and expression 2 can be any expressions; the set of keywords forming the allowed comparators includes:

 (LESSTHAN , GREATERTHAN , EQUALS , NOTEQUALS , > , < , >= , <= )

Syntax

IF <condition> THEN <true-expression> ELSE <false-expression>

or

IF <condition1> <conditional operator> <condition 2> THEN <true-expression> ELSE <false-expression>

Example

Bonus = IF Sales >= Sales Level THEN Sales * .05 ELSE 0

Bonus2 = IF Sales >= Sales Level AND Accounts >= Accounts Level THEN Sales * .05 ELSE 0

IN

The IN keyword allows you to refer to a variable that exists outside of the current node for uses in equations.

Syntax

<variable name> IN <node name>

Example

“Node A” has equations:

b = 2

q = 8

“Node B” has equation:

a = 5

b = 6

Total = 20 + a * (b + (q / 4)) IN Node A

IRR

The IRR (Internal Rate of Return) function calculates the discount rate at which the Net Present Value of a given set of cash inflows and cash outflows equals zero. If the IRR is a negative value or greater than 200%, the application will return a value of 'NaN'1 for the IRR. Similar to NPV, Initial Investment and Cash Flow are used here in a generic sense and can be referred to by the other variable names. Numerical data cannot be entered directly into the IRR function -- only variables can be used in the equation. Hence, myIRR = IRR (10000, 2200) is NOT an acceptable use of the IRR function.

Syntax

IRR (Initial Investment, Cash Flow)

Example

Initial investment = 100000, 0, 0, 0

Cash in= 0, 45000, 76000, 108000

Cash out = 0, 15000, 26000, 28000

Cash flow = Cash in - Cash out

Internal rate of return = IRR (Initial investment, Cash flow)

[RESULTS] = NaN, NaN, NaN, 0.23

Note: NaN is an abbreviation for 'Not a Number'.
LN

The LN function calculates the Natural Log of a number.

Syntax

LN (<variable name>)

LN (<number>)

Example

Numbers = 17, 27, 33

Natural log 1 = LN (Numbers)

[RESULTS] = 2.83, 3.3, 3.5

Natural log 2 = LN (56)

[RESULTS] = 4.03

MATRIX

The MATRIX function can be used to return the value of the selected variable in a specified column. The column number is the order number in which the column appears in the column statement.

Syntax

MATRIX (<variable name>, <column number>)

Example

Column Statement: 2006 - 2010

Sales = 2000, 4000, 6000

Profit = 50000, 20 * MATRIX(Sales, 2), 30 * MATRIX(Sales, 3)

(MATRIX (Sales, 2) retrieves the value for sales in the second column which is 2007)

[RESULTS] = 50000, 80000, 180000

MAX

The MAX function calculates the maximum value of the expressions in a given set of expressions. Expressions can be variable names, numbers, arithmetic expressions, other functions or any combinations of them. 

Syntax (Used in Special Column Window or Equation Window)

MAX (<expression1>, <expression2>, …)

Example

Y = 100

X = 50

Max = MAX(60, X, Y)

[RESULTS] = 100

MIN

The MIN function calculates the minimum value of the expressions in a given set of expressions. Expressions can be variable names, numbers, arithmetic expressions, other functions or any combinations of them. The usage of the function must have the following format:

Syntax(Used in Special Column Window or Equation Window)

MIN (<expression1>, <expression2>, …)

Example

Y = 100

X = 50

Min = MIN ( 60, X, Y)

[RESULTS] = 50

MOVAVG

The MOVAVG function calculates the moving average.

Syntax

MOVAVG (variable name)

Example

Sales = 200, 400, 600

Avg Sales = MOVAVG(Sales)

[RESULTS] = 200, 300, 400

NORAND

The NORAND function in used in Monte Carlo Simulations. The results are used for the Risk Analysis Chart. For all other results the mean value is used.

Syntax

NORAND ( <mean>, <standard deviation> )

Example

Units Sold = NORAND(200, 275)

NPV

The Net Present Value (NPV) of a capital budgeting project indicates the dollar value above the discount rate for a project at a particular point in time. A variable using this keyword must use it like a function call and must be of the following format: Initial Investment, Cash Flow, and Discount Rate values are used in a generic sense and need to be replaced by variable names that contain the data to be used as cash flow. It is also important to note that numerical data cannot be entered directly into the IRR function -- only variables can be used in the equation. Hence, myNPV = NPV (300.50, 229, 0.15) is NOT an acceptable use of the NPV function.

Syntax

NPV (Init Investment, Cash Flow, Discount Rate).

Example

Initial investment = 100000, 0, 0, 0

Cash in= 0, 45000, 76000, 108000

Cash out = 0, 15000, 26000, 28000

Cash flow = Cash in - Cash out

Discount rate = .15

Net present value = NPV (Initial investment, Cash flow, Discount rate)

[RESULTS] = -86,956.52, -64,272.21, -31,396.4, 14.343.86

PERIOD

The PERIODS keyword is used to define how many columns make up one year.

Syntax

<columns>

PERIODS <number of periods per year>

Example

quarter 1 - quarter 8

PERIODS 4

Note: By default the functions NPV, IRR, and BCRATIO assume that the columns declared in the Column statement are years. If the columns do not represent years, the PERIODS statement must be used to define the number of columns that make up one year.

POWER

The POWER function takes a base value and multiplies it by itself the number of time specified by the exponent.

Syntax

POWER ( <base value>, <exponent>)

Example

Four squared = POWER(4, 2)

[RESULTS] = 16

PREVIOUS

The PREVIOUS keyword is used in an equation when the value of the variable in one column is derived from that of a previous column. A column offset specifies how far back to retrieve a previous value. If no column offset is specified then 1 is used. If PREVIOUS is used with a variable name after it, that variable’s value in the pervious column will be used. If no value is available then zero will be used.

Syntax

PREVIOUS <column offset> <variable name>

Example

Sales = 17000, 20000, PREVIOUS * 1.07

OR

Sales = 17000, 20000, PREVIOUS 2 * 1.07

OR

Sale ratio = 5, 6, 7

Sales = 10000, PREVIOUS Sale ratio * 2000

OR

Sale ratio = 5, 6, 7

Sales = 10000, PREVIOUS 2 Sale ratio * 2000

ROUND

The ROUND function returns the closest whole number. In other words, this function takes the given variable and rounds it up or down depending on which is closest to the nearest whole number. If the number is equally between two whole numbers then it is rounded up.

Syntax

ROUND (<variable name>) (Used in Special Columns or Equations)

ROUND (<variable name>, <decimal place>) (Used in Equations)

Example

numbers = 1.7, 2.3, 4.5, 6.4

round = ROUND (numbers)

[RESULTS] = 2, 2, 5, 6

STDEV

The STDEV function calculates the standard deviation of the specified columns. The variable name is the variable for which standard deviation value is to be calculated. The column number represents the starting and ending columns from which the values for variable name will be selected to calculate the standard deviation.

Syntax

STDEV (<variable name>,<start column num>,<end column num>)

Example

standardDev = STDEV(Units sold, 1, 2)

SUM

The SUM function is used to calculate the sum of the values of specified variables. The SUM function uses the THRU keyword to specify the range of variables or columns to be used in it.

Syntax

SUM (variable 1 THRU variable 2)

SUM (COL <column name> THRU COL <column name>)

Example

a = 2

b = 4

c = 7

d = SUM (a THRU c)

Example (Special Column)

column = SUM(COL 2006 THRU COL 2009)

TOTAL

The TOTAL function is used to calculate the total of the values of all instances of a variable in the model or in a range of nodes in the model. The TOTAL function uses the THRU keyword to specify the range of variables to be used in it.

Syntax

TOTAL ( <variable name> )

TOTAL ( <variable name> IN <start node> THRU <end node> )

Example

“Node A” has equations:

a = 2

b = 8

“Node B” has equations:

a = 5

b = 6

“Node C” has equations:

a = 5

b = 6

c = TOTAL(a)

d = TOTAL(b IN Node B THRU Node C)

TREND
The TREND keyword creates a trend line from previous columns values and projects those values into future columns using linear regression.

There mus be at least 2 columns of previous information in order for TREND to be used

Syntax

val,val, TREND
Example

Sales = 100, 145, 170, 220, TREND
TRIRAND

The TRIRAND function in used in Monte Carlo Simulations. The results are used for the Risk Analysis Chart. For all other results the most likely values are used.

Syntax

TRIRAND ( <min possible>, <most likely>, <max possible> )

Example

Units Sold = TRIRAND(200, 275, 345)
